Visual Basic for Applications

Module 2

Introduction

The purpose of this module is to introduce automation of Microsoft Office products.  By the end of this module you should be able to accomplish the following tasks:

· Automate one Office product from within a different Office product

· Understand how to reference the Object Model of different Office products

· Use debugging techniques

· Implement error handling



Automation

There is a certain amount of automation you can do within Microsoft Office.  Automation means to run commands in an Office product without the need of user assistance.  Microsoft has said that automation is not supported in Office products and that they are designed to be user driven; however, it still can be done very efficiently.

Automating Microsoft Word

In this example we will automate Microsoft Word from within Microsoft Excel.  A customer has called with an urgent request.  She received a Microsoft Excel spreadsheet that has names and phone numbers listed in rows but some columns have the name and some have the phone number.  The data is not consistent.  She needs to have the data imported into a Word table.  The Excel file will be emailed to you.

Getting Started

To begin with we need a way to start Microsoft Word from within Excel.  To do this we add a reference to the Microsoft Word Object Library.  Open DataFile.xls and enter the VBE.  Make sure you have something within DataFile.xls selected in the Project Explorer before doing the next step.  Click Tools, References and choose Microsoft Word 10.0 Object Library (or whatever version # you have.)  This now gives us access to Word’s Object Model as if we were in Word.

Insert a module into DataFile.xls.  Next type the following lines of code:

1
Public Sub AutomateWord()

2
Dim wdApp As Word.Application

3
Dim wdDoc As Word.Document

4


5
Set wdApp = New Word.Application

6
wdApp.Visible = True

7

8
Set wdDoc = wdApp.Documents.Add

9

10
Dim TotalRows As Integer, i As Integer

11
TotalRows = 1

12
While Cells(TotalRows, 1) <> ""

13
    TotalRows = TotalRows + 1

14
Wend

15

16
TotalRows = TotalRows - 1

17

18
wdDoc.Tables.Add wdApp.Selection.Range, TotalRows, 2, ,  
wdAutoFitFixed

19

20
For i = 1 To TotalRows

21
    With wdApp.Selection

22
        If IsNumeric(Cells(i, 1)) Then

23
            .TypeText Cells(i, 2)

24
            .MoveRight wdCell

25
            .TypeText Cells(i, 1)

26
            .MoveRight wdCell

27
        Else

28
            .TypeText Cells(i, 1)

29
            .MoveRight wdCell

30
            .TypeText Cells(i, 2)

31
            .MoveRight wdCell

32
        End If

33
    End With

34
Next i

35
End Sub


Line 2 declares a variable that will hold a handle to an instance of Microsoft Word.  

Line 3 declares a variable that will be used as our handle to a Word document.  

Line 5 actually starts a new instance of Microsoft Word and if you don’t do line 6 you will not see Word.

Line 8 creates a new document and passes the handle to our variable.

Lines 10-16 are used to determine how many rows our new table in Word will need.

Line 18 adds a new table to our Word document.

Line 20 starts a loop to go through each row in Excel.

Line 21 uses a With statement.  “With” is shortcut syntax.  In this case we do not have to keep typing wdApp.Selection. 

Line 22 tests to see if the cell in the current row and column 1 is a number.  If it is we assume it to be the phone number and therefore column 2 in Excel must have the name.  We put the name in the table cell, move to the right and put in the phone number.

Lines 22-32 contain the logic to decide which cell has the phone number so that it can be put in the proper location in the Word table.

Run the code and see what happens.

Exercise 1
Modify the above example so that the phone numbers in the DataFile.xls are not just extensions but have the format ###.###.#### ( i.e. 770.343.0680).  You will need to change the logic in lines 22-32.  The IsNumeric function will not work in this case.  (Hint:  There are several ways to do it.  Check Help on using the Instr function.)  Add code to save the new word document after the import is done.


Debugging

Debugging is the process of getting rid of bugs in your code and making it stable.  In the VBE there are different debugging options under the Debug menu.  I will highlight some of the key debugging features.

Step Into

Step Into is found under the debug menu and F8 is the keyboard equivalent.  This command allows you to step line by line through your code.  This is an extremely powerful tool for narrowing down where a bug may be.

Watch

A watch is used when you want to know when the value of a certain variable or expression either changes or is at a certain value.  For example, if you have code that calculates salaries and for a certain user the salary results in $100,000,000,000.00 (and it is not Bill Gates) you can add a watch on your salary variable.  Under Debug choose Add Watch and tell the compiler to stop when salary is greater than $70,000.00.  You can then run your code and when the salary gets over $70,000 the code will enter debug mode and you can see what line of code caused the error.
Breakpoints

Breakpoints are used to stop code execution at a certain point.  You can click the left gray margin in the VBE to set a breakpoint, press F9, or use the Debug menu.  When you run your code and it gets to a breakpoint the code will stop at the line without running it.  You can then use other debugging options to continue.  This is very useful when you have a lengthy routine located before the line of code where you know the error is.

Immediate Window
The immediate window can be shown by pressing Ctrl  + G or under the View menu.  In the immediate window you can run lines of code or view certain variables.  For example, you can type ?Now in the immediate window and it will print the current time.  ? or Print and then a statement or a variable will show the result.  You can also run lines of code.


Error Handling
Error handling means what it says, to handle errors.  Most errors to well written code are caused by external factors such as: network drive down, no floppy in drive when saving to A drive, user types in wrong data, etc.  You should implement error handling to prevent your code from crashing if an unexpected error occurs.

This next sample illustrates one method of error handling.  Open Microsoft Excel and enter the VBE.  Type the following code into a new module:
1
Public Sub ErrorHandlingSample()

2
On Error GoTo ErrHandler

3
Dim r As String

4

5
ActiveCell.Value = 5 / 0
6
ActiveCell.Value = 1 + r

7
ActiveCell.Value = 1

8

9
Exit Sub

10
ErrHandler:

11
    If Err.Number = 11 Then

12
        MsgBox "You cannot divide a number by 0."

13
    Else

14
        MsgBox Err.Number & ": " & Err.Description

15
        Resume Next

16
    End If

17
End Sub

Step through this code by using the F8 key to see exactly what happens and in what order.

Line 2 begins our error handler.  It says that when an error occurs, go to the line we have called ErrHandler.
Line 5 will generate a divide by zero error.  You will notice that our custom error message is displayed.  Fix the error by replacing the 0 with some other number.

Line 6 will now generate an error because it tries to add 1 to a string.  You will notice that a generic error will be shown and then the code will continue at the next line.

Line 9 exits the sub otherwise code would continue into the error handler even if there was no error.
Line 10 is the label for our error handler.

Line 11 tests first for a known possible error and if it isn’t that one it presents a generic error message.

Line 15 causes the code to resume at the next statement after the one that caused the error.  If you handle an error you can call Resume and that will rerun to the line of code that caused the error.  If you call Resume and have not handled the error you will cause an infinite loop.


Summary

To automate one Office product from within a different one you should first set a reference to it from Tools, References.  This gives you access to that product’s object model.

It is important to implement error handling since you never know what types of errors will occur.  Different debugging techniques allow you to better implement your error handling by narrowing down where errors may occur and also finding out the error numbers for specific errors.

Module 3 will incorporate the lessons learned here with more real life examples.

